

Modern Assembly Language Programming  
with the  
ARM processor  
Chapter 6: Abstract Data Types

## Contents

## ADT

An Abstract Data Type (ADT) is composed of:

- data and
- the operations that work on that data.

Separate the *interface* from the *implementation*!

Object Oriented Programming (OOP) is a related concept with similar goals.

- C provides support for ADTs and allows the implementation to be done in assembly.
- C++ provides support for objects and allows the implementation to be done in assembly (but you need to know a bit more about the compiler).

## Information Hiding

- Hide the physical storage layout for data.
- If the data structure changes, it only affects a small subset of the total program
  - Unit Testing
  - Rapid Prototyping
  - Standardization
  - Debugging
  - Code Reuse
  - Independent Development
  - Larger Projects
  - Higher Quality

## ADT Interface in C

```
1 #ifndef IMAGE_H
2 #define IMAGE_H
3 #include <stdio.h>
4
5 typedef unsigned char pval;
6
7 struct imageStruct;
8
9 typedef struct imageStruct Image;
10
11 Image *allocateImage();
12 void freeImage(Image *image);
13
14 int readImage(FILE *f, Image *image);
15 int writeImage(FILE *f, Image *image);
16
17 int setPixelRGB(Image *image, int row, int col, pval r, pval g, pval b);
18 int setPixelGray(Image *image, int row, int col, pval v);
19
20 pval getPixelRGB(Image *image, int row, int col);
21 pval getPixelGray(Image *image, int row, int col);
22
23 #endif
```

## ADT Private Declarations

A separate header, only available to the Image implementation files:

```
1 #ifndef IMAGE_PRIVATE_H
2 #define IMAGE_PRIVATE_H
3 #include <image.h>
4
5 typedef struct {
6     pval r,g,b;
7 } Pixel;
8
9 struct imageStruct;
10 int rows;      // number of rows in the image
11 int cols;      // number of columns in the image
12 Pixel *pixels; // array of pixel data
13 };
14 #endif
```

## ADT Private Declarations

imagedefs.S

```
1  @@@ Definitions for pixel and image data structures
2
3      @@ pixel
4      .equ    p_red,    0  @ offset to red value
5      .equ    p_green,  1  @ offset to green value
6      .equ    p_blue,   2  @ offset to blue value
7      .equ    p_size,   3  @ size of the pixel data structure
8
9      @@ image
10     .equ    i_rows,   0  @ offset to number of rows
11     .equ    i_cols,   4  @ offset to number of columns
12     .equ    i_pixels, 8  @ offset to pointer to image data
13     .equ    i_size,  12  @ size of the image data structure
```

## ADT Implementation in Assembly

```
1      .include "imagedefs.S" @ include data layout
2      .text
3      /* Implementation of
4      int setPixelRGB(Image *image,int row,int col,pval r,pval g,pval b);
5      */
6      .global setPixelRGB
7      setPixelRGB:
8          stmfd    sp!, {r4,r5} @ We will need some registers
9          @@ Calculate address of desired pixel
10         ldr      r4, [r0, #i_cols] @ get # columns in image
11         mla      r4, r4, r1, r2 @ calculate offset to desired row
12                                     @ and add offset to desired column
13         mov      r5, #p_size @ load size of a pixel
14         mul      r4, r4, r5 @ get offset to desired pixel in bytes
15         ldr      r5, [r0, #i_pixels] @ get pointer to the pixel array
16         add      r0, r4, r5 @ add offset to desired pixel
17         @@ Set the rgb fields in the pixel
18         ldr      r4, [sp, #8] @ load g from stack
19         ldr      r5, [sp, #12] @ load b from stack
20         strb     r3, [r0, #p_red] @ store r,g,b in pixel struct
21         strb     r4, [r0, #p_blue]
22         strb     r5, [r0, #p_green]
23         ldmfd    sp!, {r4,r5}
24         mov      pc, lr
```

## Faster ADT Implementation in Assembly

We know that a pixel is three bytes, so replace the multiply with a shift/add operation.

```
1      .include "imagedefs.S" @ include data layout
2      .text
3      /* Implementation of
4      int setPixelRGB(Image *image,int row,int col,pval r,pval g,pval b);
5      */
6      .global setPixelRGB
7      setPixelRGB:
8          stmfd    sp!, {r4,r5} @ We will need some registers
9          @@ Calculate address of desired pixel
10         ldr      r4, [r0, #i_cols] @ get # columns in image
11         mla      r4, r4, r1, r2 @ calculate offset to desired row
12                                     @ and add offset to desired column
13         add      r4, r4, r4, lsl #1 @ multiply by 3
14         ldr      r5, [r0, #i_pixels] @ get pointer to the pixel array
15         add      r0, r4, r5 @ add offset to desired pixel
16         @@ Set the rgb fields in the pixel
17         ldr      r4, [sp, #8] @ load g from stack
18         ldr      r5, [sp, #12] @ load b from stack
19         strb     r3, [r0, #p_red] @ store r,g,b in pixel struct
20         strb     r4, [r0, #p_blue]
21         strb     r5, [r0, #p_green]
22         ldmfd    sp!, {r4,r5}
23         mov      pc, lr
```

## Word Frequency Counts

Counting the frequency of words in written text has several uses.

- Digital forensics: provide evidence as to the author of written communications.
- Document Classification: suggest similar books or help with sorting documents for storage and access.

Many ways to implement this ADT!

- Linked List
- Binary Tree
- Trie
- ...

## Wordlist Interface in C

```
1 #ifndef LIST_H
2 #define LIST_H
3
4 /* Define an opaque type, named wordlist */
5 typedef struct wlist wordlist;
6
7 /* wl_alloc allocates and initializes a new word list. */
8 wordlist* wl_alloc();
9
10 /* wl_free frees all the storage used by a wordlist */
11 void wl_free(wordlist* wl);
12
13 /* wl_increment adds one to the count of the given word. */
14 /* If the word is not in the list, then it is added with */
15 /* a count of one. */
16 void wl_increment(wordlist *list, char *word);
17
18 /* wl_print prints a table showing the number */
19 /* of occurrences for each word, followed by the word. */
20 void wl_print(wordlist *list);
21
22 /* wl_print_numerical prints a table showing the number */
23 /* of occurrences for each word, followed by the word, */
24 /* sorted in reverse order of occurrence. */
25 void wl_print_numerical(wordlist *list);
26
27 #endif
```

## Wordlist Private Declarations

### C Data structures:

```
1  /* The wordlistnode type is a linked list of words and      */
2  /* the number of times each word has occurred.              */
3  typedef struct wlist_node{
4     char *word;
5     int count;
6     struct wlist_node *next;
7 }wordlistnode;
8
9  /* The wordlist type holds a pointer to the linked list     */
10 /* and keeps track of the number of words in the list      */
11 typedef struct wlist{
12     int nwords;
13     wordlistnode *head;
14 }wordlist;
```

### Assembly definition:

```
1  @@@ Definitions for the wordlistnode type
2     .equ     wln_word,0      @ word field
3     .equ     wln_count,4    @ count field
4     .equ     wln_next,8     @ next field
5     .equ     wln_size,12    @ sizeof(wordlistnode)
6  @@@ Definitions for the wordlist type
7     .equ     wl_nwords,0    @ number of words in list
8     .equ     wl_head,4     @ head of linked list
9     .equ     wl_size,8     @ sizeof(wordlist)
```

## Problems with Wordlist Implementation

### Linked List:

- Inserting or updating an item in the list requires  $\frac{n}{2}$  comparisons on average to find the insertion point. It is an  $O(n)$  algorithm. It is very slow for large text inputs.
- Re-sorting the list by frequency is also very slow.

### Binary Tree?

- Inserting or updating an item in the tree requires  $\log_2 n$  comparisons on average. It is an  $O(\log n)$  algorithm.
- It is much faster than a linked list.

Worst case, there are about 1,000,000 words in the English language. 500,000 comparisons for linked list versus 19.9 comparisons for tree. Speedup of  $\approx 25125$ .


## Problems with Wordlist Tree-based Implementation

Re-sorting the list by frequency could still be slow.

**Solution:** Add an index, then sort the index with quicksort!


The index is just an array of pointers; one for each node in the tree.

## Binary Tree of Word Frequencies




## Binary Tree of Word Frequencies with Index

The index was initialized using an in-order traversal of the tree.



## Binary Tree of Word Frequencies with Index

The index was sorted using a quicksort.



## Wordlist Private Declarations: Tree Version

These changes are not visible to any program using this ADT.

### C Data structures:

```
1 /* The wordlistnode type is a binary tree of words and */
2 /* the number of times each word has occurred. */
3 typedef struct wlist_node{
4     char *word;
5     int count;
6     struct wlist_node *left, *right;
7     int height;
8 }wordlistnode;
```

### Assembly definition:

```
1 @@@ Definitions for the wordlistnode type
2     .equ    wln_word,0      @ word field
3     .equ    wln_count,4    @ count field
4     .equ    wln_left,8     @ left child pointer
5     .equ    wln_right,12   @ right child pointer
6     .equ    wln_height,16  @ height of this node
7     .equ    wln_size,20    @ sizeof(wordlistnode)
```

## Printing in Order of Frequency part 1

```
1  @@@ -----
2  @@@ wl_print_numerical prints a table showing the number
3  @@@ of occurrences for each word, followed by the word,
4  @@@ sorted in reverse order of occurrence.
5  .global wl_print_numerical
6  wl_print_numerical:
7      stmfid    sp!, {r4-r6,lr}    @ save registers
8      mov      r4,r0              @ copy original pointer
9      ldr      r5,[r0,#wl_nwords]@ load nwords
10     lsl      r0,r5,#2           @ multiply by four (size of pointer)
11     bl       malloc             @ allocate storage for the index
12     cmp      r0,#0              @ check return value
13     bne     malloc_ok
14     ldr      r0,=failstr        @ load pointer to string
15     bl       printf
16     mov      r0,#1
17     bl       exit              @ exit(1)
18 malloc_ok:
```

## Printing in Order of Frequency part 2

```
1 malloc_ok:
2     mov     r6,r0           @ save pointer to array
3     ldr     r1,[r4,#wl_head]@ get pointer to tree
4     bl     getptrs        @ fill in the pointers
5     mov     r0,r6           @ get pointer to array
6     add     r1,r0,r5,lsl #2 @ get pointer to end of array
7     sub     r1,r1,#4
8     bl     wl_quicksort   @ re-sort the array of pointers
9     @@ Print the word frequency list.
10    mov     r4,#0           @ do a for loop
11 loop:   cmp     r4,r5
12         bge     done
13         ldr     r0,=fmtstr
14         ldr     r3,[r6,r4,lsl #2] @ get next pointer
15         add     r4,r4,#1
16         ldr     r1,[r3,#wln_count]@load count
17         ldr     r2,[r3,#wln_word] @load ptr to word
18         bl     printf
19         b     loop
20 done:   ldmfd  sp!,{r4-r6,pc} @ restore & return
```

## Initializing the Array of Pointers (the index) part 2

```
1  @@@ -----  
2  @@@ wordlistnode **getptrs(wordlistnode *ptrs[],wordlistnode *node)  
3  @@@ this function recursively traverses the tree, filling in the  
4  @@@ array of pointers.  
5  @@@ r0 is incremented as each pointer is stored, so it returns  
6  @@@ a pointer to the next pointer in the array that needs to  
7  @@@ be set.  
8  getptrs:  
9      cmp     r1,#NULL           @ if node == NULL  
10     moveq   pc,lr             @ return immediately  
11     stmfd   sp!,{r4,lr}  
12     mov     r4,r1             @ save address of current node  
13     ldr     r1,[r4,#wln_left] @ get ptr to left child  
14     bl     getptrs           @ process left child  
15     str     r4,[r0],#4        @ Store address of current node  
16     ldr     r1,[r4,#wln_right]@ get ptr to right child  
17     bl     getptrs           @ process right child  
18     ldmfd   sp!,{r4,pc}
```

## Quicksort on Array of Pointers (the index) in C

```
1 wl_quicksort(wordlistnode **left,wordlistnode **right)
2 {
3     wordlistnode **first=left, **last=right;
4     wordlistnode *tmp;
5     int pivot;
6     if(left < right)
7     {
8         pivot=(*left)->count;
9         do{
10            while((left <= right) && ((*left)->count > pivot))
11                left++;
12            while((left <= right) && ((*right)->count < pivot))
13                right--;
14            if( left <= right )
15                {
16                    tmp = *left;
17                    *left = *right;
18                    *right = tmp;
19                    left++;
20                    right--;
21                }
22        }while(left<=right);
23        wl_quicksort(first,right);
24        wl_quicksort(left,last);
25    }
26 }
```

## Quicksort on Array of Pointers (the index) part 1

```
1  @@@ -----  
2  @@@ function wl_quicksort(wln **left,wln **right) quicksorts  
3  @@@ the array of pointers in order of the word counts  
4  wl_quicksort:  
5      cmp     r0,r1  
6      movge  pc,lr           @ return if length<=1  
7      stmfd  sp!,{r4-r7,lr}  
8      ldr   r12,[r0]        @ use count of first item as  
9      ldr   r12,[r12,#wln_count] @ pivot value in r12  
10     mov   r4,r0           @ copy current left  
11     mov   r5,r1           @ copy current right  
12     mov   r6,r0           @ original left(first)  
13     mov   r7,r1           @ original right(last)  
14 loopa:  cmp   r4,r5       @ while ((left <= right) &&  
15         bgt   finale  
16         ldr   r0,[r4]     @ ((*left)->count > pivot))  
17         ldr   r1,[r0,#wln_count]  
18         cmp   r1,r12  
19         ble  loopb  
20         add  r4,r4,#4     @ increment left  
21         b    loopa
```

## Quicksort on Array of Pointers (the index) part 2

```
1 loopb:  cmp     r4,r5           @ while left <= right &&
2         bgt     finale
3         ldr     r2,[r5]       @ (*right)->count < pivot
4         ldr     r3,[r2,#wln_count]
5         cmp     r3,r12
6         bge     cmp
7         sub     r5,r5,#4      @ decrement right
8         b       loopb
9 cmp:    cmp     r4,r5           @ if( left <= right )
10        bgt     finale
11        str     r0,[r5],#-4   @ swap pointers and
12        str     r2,[r4],#4    @ change indices
13        b       loopa
14 finale: mov     r0,r6           @ quicksort array from
15        mov     r1,r5           @ first to current right
16        bl     wl_quicksort
17        mov     r0,r4           @ quicksort array from
18        mov     r1,r7           @ current left to last)
19        bl     wl_quicksort
20        ldmfd  sp!,{r4-r7,pc}
```

## Therac-25

The Therac-25 was a device designed for radiation treatment of cancer.

Although this machine was built with the goal of saving lives, between 1985 and 1986, three deaths and other injuries were attributed to the hardware and software design of this machine.

## Therac-25

The worst problems in the design and engineering of the machine were:

- The code was not subjected to independent review.
- The software design was not considered during the assessment of how the machine could fail or malfunction.
- The operator could ignore malfunctions and cause the machine to proceed with treatment.
- The hardware and software were designed separately and not tested as a complete system until the unit was assembled at the hospitals where it was to be used.
- The design of the earlier Therac-6 and Therac-20 machines included hardware interlocks which would ensure that the X-ray mode could not be activated unless the tungsten radiation shield was in place. The hardware interlock was replaced with a software interlock in the Therac-25.
- Errors were displayed as numeric codes, and there was no indication of the severity of the error condition.

## Therac-25

**Although the code was written in assembly language, that fact was not cited as an important factor.**

The fundamental problems were

- poor software design,
- over-confidence, and
- re-use of code in an application for which it was not initially designed.

A proper design using established software design principles, including structured programming and abstract data types would almost certainly have avoided these fatalities.